
連分数と連分数展開による関数計算の実際 

(1) 連分数とは 

連分数（continued fraction）とは，以下のように分母にさらに分数が含

まれている分数です。 

 

 

 

 

連分数のうち，分子がすべて 1（上の例では，b0 = b1 = b2 = 1）のものを，

特に正則連分数（regular continued fraction）といいます。単に連分数と

いうとき，正則連分数を指すことが多いようです。正則連分数の具体例を以

下に示します。 

 

 

 

 

(2) 連分数の表記法のバラエティ 

連分数表記は，(1)の方法が直感的に分かりやすいのですが，縦方向にど

んどん伸びてしまいますので，次のように略記する方法もあります。 

 

 

または，正則連分数であることを前提として， 

 

 

と表記する方法もあります。さらに，級数と同様，無限に続く連分数も考え

ることができます。 

(3) 連分数の計算方法 

ある数 x が与えられたとき，x を超えない最大の整数を a0とし， 

 

 

を満足するよう x1を設定します。さらに，x1が整数でなければ，x1を超え

ない最大の整数を a1とし， 

 

 

を満足するよう x2を設定します。以下，これを繰り返して，連分数 

 

 

 

 

 

 

を得ることができます。連分数の数字列（文字列）に変換するプログラム例

を以下に示しますので，色々ためしてみましょう。なお，後が乱れますが，

誤差のせいですのでご勘弁ください。 

 

【プログラム例】 
Public Function contFrac(X) As String 
  Dim S As String: B = Fix(X) 
  If Abs(B - X) < 0.0000001 Then 
    contFrac = B 
    Exit Function 
  End If 
  S = B 
  For i = 1 To 20 
    X = 1 / (X - B): B = Fix(X): S = S & " : " & B 
  Next 
  contFrac = S 
End Function 

(4) 連分数の例 

A. 2 次方程式の解としての連分数 

今，次のような 2 次方程式を考えてみましょう。 

 

 

この式を変形すると， 

 

 

 

 

すなわち，      は，              の解のうち正符号の解であるこ

とが分かります。すなわち， 

 

とすることができます。 

 

B. 黄金数（golden number）φ 

連分数の世界では，特に       の解を黄金数といいます。なお，

復活祭日を決めるためにキリスト教圏で用いられる数字，すなわち西暦年数

に 1 を加えて 19 で割った値も黄金数（golden number）といいますが，こ

れとは別物です。 

変形すると， 

 

 

 

 

すなわち，黄金数         となります。  

          

C. 平方根の計算 

         の解が       であることは，解の公式から明ら

かです。今，連分数を 

の形で表すと， 

 

 

 

 

 
 

一方， 
 
ですから            となります。したがって 

 

 

を得ることができます。たとえば，n = 1, m = 2 のとき，すなわち 

   

  

となるのはこのためです。 

 本来の定義では，連分数の各係数は整数ですが，整数にこだわらなければ，

このことを使って平方根を求めることができます。すなわち， 

    

 

とすることで平方根を求めることができます。以下にそのプログラミング例

を VB で示します。 

 

【プログラム例】 
Public Function mySqr(X) 
   XX = Abs(X) 
   If XX < 0.00000000001 Then 
       mySqr = 0: Exit Function 
   End If 
   less1 = False    '1 より小さい時，逆数をとる 
   If XX < 1 Then  
     XX = 1 / XX: less1 = True 
   End If 
   N = 1: NN = 1  '平方が XX を超えない整数を求める 
   Do While (NN < XX) 
     N = N + 1: NN = N * N 

   Loop 
   If Abs(NN - XX) < 0.00000000001 Then 
       mySqr = N: Exit Function 
   End If 
   N = N - 1: NN = N * N 
   DX = XX - NN 
   If Abs(DX) < 0.00000000001 Then 
       mySqr = N: Exit Function 
   End If 
   DN = 2 * N: DM = DN / DX   '連分数の計算 
   Dim Z As Double: Z = 0 
   For i = 1 To 20 
      Z = 1 / (DM + 1 / (DN + Z)) 
   Next 
   mySqr = N + Z   '元データが 1より小さい時，逆数が結果 
   If less1 Then mySqr = 1 / mySqr 
End Function 
 

D. 白銀数 

      を白銀数といいます。その逆数は， 

 

 
 

となります。 
 
E. 円周率πの値 

円周率の正則連分数には，以下のように規則性がないと考えられています。 

 
 

ただ，正則でない連分数では規則性を持つ場合があります。 

 

 

 

 
 
円周率を計算するには，右のほうが早く収束するようです。たとえば，

Excel VBA で次のような関数を作成して確認してみましょう。引数の N は

繰返し回数です。N を変えて比較すると収束の様子が分かります。 

収束の様子をグラフ化してみました。上図が左の連分数，下図が右の連分数

です。 
 
 
 

 
 
 
 
 
 
 
 
 
 
 
 
 
 

F. ネイピア数 e（自然対数の底） 

ネイピア数の連分数展開は循環しませんが，規則性を持っています。すな

わち，以下のようになります。 

 

 

整数部以外は，1, 2n, 1 (n = 1, 2, ・・・, ∞) が続いています。これを利用し

てネイピア数を計算することができます。 

【左の連分数】 
Public Function myPi1(N) 
  NN = N * 2 + 1: A = 7 
  For i = NN To 1 Step -2 
    A = 6 + (i * i) / A 
  Next 
  myPi1 = A - 3 
End Function 

【右の連分数】 
Public Function myPi2(N)' 
  A = 2 * N 
  For i = N To 1 Step -1 
    A = 2 * i - 1 + (i * i) / A
  Next 
  myPi2 = 4 / A 
End Function 

],14,1,1,12,1,1,10,1,1,8,1,1,4,1,1,2,1;2[ e

左の連分数 

右の連分数 

【プログラム例】 
Public Function myExp(N) As Double 
  Dim A As Double 
  NN = N * 2: A = 1 
  For i = NN To 2 Step -2 
    A = 1 + 1 / (i + 1 / (1 + 1 / A)) 
  Next 
  myExp = 2 + 1 / A 
End Function 

 
 
 
 
 
 
 
 
 
 
 
 

 

(5) 連分数の性質 

A. 数列 pn, qnの定義 

いま，正の整数 an (n = 0, 1, 2, 3,・・・)の数列，a0, a1, a2, a3,・・・を考え，

数列 pn, qnを 

 

 

とすると，連分数は 

 

 

となることが知られています。 

 

B. ユークリッドの互除法との関係 

数列 pnの定義を逆に読むと「被除数 pnを除数 pn-1 で除した商を an-1,とし，

新しい被除数を pn-1とし，除数を余り pn-2 とする」と読むことができます。

すなわち，n の方向を逆に辿ればユークリッドの互除法を計算し，その過程

で現れた商が連分数の数列となっているわけです。いわば，上記数列の定義

は互除法の操作を逆に辿っていると考えることができます。 

さらに，pn, qn は整数であり，かつユークリッドの互除法を適用していま

すので，pnと qnは互いに素の関係にあります。すなわち，pn /qn は，すでに

通分された関係ですから，既約分数となります。 

 

C. pnと qnの関係 

次のような計算を行ってみましょう。 

 

 

 

 

いま，           が成り立つとすれば， 

 

 

 

 

が成り立ちますので，数学的帰納法により 

 

 

が成立します。 

 

D. 黄金数のとき 

数列 anがすべて１のとき，数列 pn, qn はフィボナッチ数列（F0 = 0, F1 = 1）

の形をしています。ですから， 

 

 

の関係が成立します。このとき，連分数は黄金数ですから，隣り合うフィボ

ナッチ数列の比は黄金数に収束することが分かります。この様子を Excel

の計算式として定義し，グラフを描いたものを以下に示します。以下のよう

に黄金数に収束している様子が分かります。 

(6) 連分数を使った関数計算 

A. 非常に収束の遅い級数の場合 

関数のマクローリン展開では，以下のように，非常に収束が遅い級数もあ

ります。 

 

 

このようなとき，前述した平方根の計算のように，関数を連分数の形で展

開して計算することで，繰返し回数が極端に減る場合があります。以下，そ

れぞれの連分数展開の形と例題としてのプログラムを示します。 

 

B. 対数（log） 

【連分数】 

 
 
【計算上の補足】 あらかじめ      となるよう調整する必要があり

ます。このため，何らかの値で除算を繰り返すことで，この範囲に収めます。

たとえば，元の値が        の場合，次のように考えて計算します。 

 

 

この場合，値 2 による除算を繰り返し，その回数分だけ log e 2 を加算しま

す。次のように考えても構いません。 

 

この場合，値 e による除算を繰り返し，その回数分だけ 1 を加算します。

ただし，値 e 自身が近似値を使わざるをえないこと，さらに値 2 による除算

の方が精度が良い等の理由で，前者の方法が若干良いようです。 

 

【プログラム例１】 
Public Function myLog(X) 
 log2 = 0.693147180559945: XX = Abs(X) 
 If XX < 0.0000000000001 Then 
   myLog = -1E+308 
   Exit Function 
 End If 
 L = 0 
 Do While XX >= 2 
   XX = XX / 2: L = L + log2 
 Loop 
 Do While XX < 1 
   XX = XX * 2: L = L - log2 
 Loop 
 XX = XX - 1: A = 0: N = 20 
 For i = N To 1 Step -1 
   A = i * XX / (2 + i * XX / (2 * i + 1 + A)) 
 Next 
 myLog = L + XX / (1 + A) 
End Function 
 

【プログラム例 2】 
Public Function myLog2(X) 
expC = 2.71828182845905: XX = Abs(X) 
 If XX < 0.0000000000001 Then 
   myLog2 = -1E+308 
   Exit Function 
 End If 
 L = 0 
 Do While XX >= 2 * expC 
   XX = XX / expC: L = L + 1 
 Loop 
 Do While XX < 1 
   XX = XX * expC: L = L - 1 
 Loop 
 XX = XX - 1: A = 0: N = 20 
 For i = N To 1 Step -1 
   A = i * XX / (2 + i * XX / (2 * i + 1 + A)) 
 Next 
 myLog2 = L + XX / (1 + A) 
End Function 

以下は，Excel に備わっている ln 関数と連分数展開のプログラムとの比

較です。ほぼ同じ値であることが分かります。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

値 e に近似値を使わざるをえないことから誤差が出ると前述しましたが，

倍精度では以下の桁数が限度です。 

e = 2.7182 81828 45905 

一方，正確な値は e = 2.7182 81828 45904 52353 … です。 

そこで，e 1= 2.7182 81828 45904，e 2= 2.7182 81828 45905，α = 0.52353 …

として，次のように計算します。 

 

 

なお，αが 1 に近いときは，次のように計算すれば構いません。 

 

 

プログラム例を以下に示します。ただし，計算機上での演算は本質的に切

り捨てです。除算結果が 0 でないとき，最終桁が切り捨てられますので，こ

の分を最後に加算しておきます。 

【プログラム例 3】 
Public Function myLog3(X) 
expc1 = 2.71828182845904: expc2 = 2.71828182845905 
 Alf = 0.523536028747135: P = Alf / (1 - Alf): PP = 1 + P 
 XX = Abs(X) 
 If XX < 0.0000000000001 Then 
   myLog3 = -1E+308 
   Exit Function 
 End If 
 L = 0 
 Do While (XX - expc2) >= expc1 
   XX = XX * PP / (expc1 + P * excp2): L = L + 1 
 Loop 
 Do While XX < 1 
   XX = XX * (expc1 + P * expc2) / PP: L = L - 1 
 Loop 
 XX = XX - 1: A = 0: N = 20 
 For i = N To 1 Step -1 
   A = i * XX / (2 + i * XX / (2 * i + 1 + A)) 
 Next 
 myLog3 = L + XX / (1 + A) 
 If myLog3 <> 0 Then myLog3 = myLog3 + 1E-16 
End Function 

 

以下のように誤差が非常に少なくなっていることが分かります。 

上記【プログラム例 1】では，log 0.8 および log 0.9 の値の誤差が大きくな

っています。これは，L = L - log 2 の部分の桁落ちのためです。log 2 の

値をより大きい桁数で示すと以下のようになりますが 

    log 2=0.6931 47180 55994 53094 17232 … 

この値が 15 桁で切られていますので，この部分を補正します。 
 
【プログラム例 4】 
Public Function myLog3(X) 
N = 20: log2 = 0.693147180559945: Flog2 = 3.09417232E-16 
 XX = Abs(X) 
 If XX < 0.0000000000001 Then 
   myLog4 = -1E+308 
   Exit Function 
 End If 
 L = 0 
 Do While XX >= 2 
   XX = XX / 2: L = L + log2 + Flog2 
 Loop 
 Do While XX < 1 
   XX = XX * 2: L = L - log2 - Flog2 
 Loop 
 XX = XX - 1: A = 0 
 For i = N To 1 Step -1 
   A = i * XX / (2 + i * XX / (2 * i + 1 + A)) 
 Next 
 myLog4 = L + XX / (1 + A) 
End Function 

C. 逆正接（arctan） 

【連分数】 
 

【計算上の補足】 まず，         の計算を考えてみると， 

①  

一方， 

 

 

 

 

 
 
 
したがって， ② 

上記①式との比較から， 

 
 

すなわち x > 1 のとき，1 / x として関数値を求め，上記計算を行います。 

一方，         のとき，同様にして， 

 

 

が求まりますので，x < 1 のとき，1 / x として関数値を求め，上記計算を行

います。 

 以上をまとめると，      でない場合，        を求め， 

 

 

 

 

とする必要があります。 

 

なお，絶対値 | x | に対して上記の計算を行い，最後に符号を調整すると

判定が少なくなります。プログラム例では，この方法を用いています。 

【プログラム例 1】 
Public Function myAtan(X) 
  N = 20: Pi2 = 3.14159265358979 / 2: XX = X: XX = Abs(X) 
  If XX <= 1 Then T = XX Else T = 1 / XX 
  A = 0 
  For i = N To 1 Step -1 
   A = (i * T) ^ 2 / (2 * i + 1 + A) 
  Next 
  TT = T / (1 + A) 
  If XX > 1 Then 
     R = Pi2 - TT 
  Else 
     R = TT 
  End If 
  If X < 0 Then R = -R 
  myAtan = R 
End Function 

 

 以下に Excel の Atan との比較を示します。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

上記【プログラム例 1】では，以下の部分で桁落ちが生じます。 
        R = Pi2 - TT 
Pi2 が近似値であるため近似値として記述した桁数より以下の少数桁の分

だけ，Rの値が小さくなっています。そこで，次のように Pi2の近似値以下

の分を加えておきます。 
 
   H = 3.23846264338328 * 0.5: HD = 0.000000000000001 '1E-15 
    ・ 
  ・ 
  ・ 

   R = Pi2 - TT + H * HD 

 

【プログラム例 2】 
Public Function myAtan2(X) 
  Pi2 = 3.14159265358979 * 0.5 
  H = 3.23846264338328 * 0.5: HD = 0.000000000000001 '1E-15 
  XX = X: XX = Abs(X) 
  If XX <= 1 Then T = XX Else T = 1 / XX 
  A = 0: N = 30 
  For i = N To 1 Step -1 
   A = (i * T) ^ 2 / (2 * i + 1 + A) 
  Next 
  TT = T / (1 + A) 
  If XX > 1 Then 
     R = Pi2 - TT + H * HD 
  Else 
     R = TT 
  End If 
  If X < 0 Then R = -R 
  myAtan2 = R 
End Function 

D. 正接（tan） 

【連分数】 
 

【計算上の補足】 特に連分数展開でなくても，マクローリン展開，または

正弦（sin）および余弦（cos）から求めることができますが，ここではあえ

て連分数展開で計算する例を示します。 

 計算する前に，あらかじめ            となるよう     分を加算

または減算します。たとえば，  より大きい場合，         と

した後，整数化して  を差し引きます。 
 
【プログラム例１】 
Public Function myTan(X) 
N = 10: Pi = 3.14159265358979: HPi = Pi / 2 

 If X > 0 Then 
    K = Fix((X + HPi) / Pi): XX = X - Pi * K 
 Else 
    K = Fix((-X - HPi) / Pi): XX = X + Pi * K 
 End If 
 AX2 = XX * XX: A = 0 
 For i = N To 1 Step -1 
   A = AX2 / ((i * 2 + 1) - A) 
 Next 
 myTan = Abs(XX) / (1 - A) 
 If XX < 0 Then myTan = -myTan 
End Function 
 

実は【プログラム例 1】には重大な欠陥があります。正の場合， 

     XX = X – Pi * K 

において，桁落ちが起きて有効桁数が少なくなります。Pi が近似値である

ため近似値として記述した桁数より以下の少数桁の分だけ，XX の値が大き

くなっています。そこで，次のように Pi の近似値以下の分を差し引きます。 

XX = X – Pi * K: XX = XX – K * 3.2385… * 1E-15  

負の場合， 

     XX = X + Pi * K 

の部分で情報落ちが発生します。ただし，倍精度浮動小数点は 15 桁ですの

で，これ以上どうしようもありません。一方， 

ですから，正の x について計算できれば構いません。そこで，次のように書

き換えることにします。 
 

【プログラム例 2】 
Public Function myTan2(X) 
 N = 10: Pi = 3.14159265358979: Hpi = Pi / 2: RPi = 1 / Pi 
 H = 3.23846264338328: DH = 0.000000000000001 ' 1E-15 
 XX = Abs(X): K = Fix((XX + Hpi) * RPi) 
 XX = XX - Pi * K: XX = XX - K * H * DH 
 AX2 = XX * XX 
 A = 0 
 For i = N To 1 Step -1 
   A = AX2 / ((i * 2 + 1) - A) 
 Next 
 myTan2 = Abs(XX) / (1 - A) 
 If XX < 0 Then myTan2 = -myTan2 
 If X < 0 Then myTan2 = -myTan2 

End Function 
 

 

myAtan より myAtan2 のほうが誤差が少なくなっています。 

E. 指数（exp） 

【連分数】 
 
ここで           までは規則的ではありません。この部分以

降が規則的な繰返しになっていることに注意してください。 
 

【計算上の補足】 指数の計算は，マクローリン展開でも十分収束が早いの

で，特に連分数展開である必要はありませんが，ここではあえて連分数で計

算する例を示します。 

あらかじめ      となるよう調整する必要があります。また 

のとき，結果の逆数をとればよいので，簡単化のために絶対値での指数を求

めます。 

   のとき，次のように考えます。 

 

すなわち，与えられた x を超えない整数 n を見つけ， e x – n を計算し，e の

整数乗 e n を乗じます。しかし，通常のコンピュータでは，機械語として乗

算は用意されますが，べき乗の命令は皆無です。一方，言語処理系では，一

般にべき乗を計算する必要がありますので，該当する計算機で初めて言語処

理系を作る場合，これをソフトウェアで実現する必要があります。 

そこで，乗算を繰り返すことになりますが，n が大きい場合，繰返し回数

が多く，非常に時間がかかります。ですから，次のように log 2 n 回のオーダ

の乗算で済ます方法が一般に用いられます。 

 
Public Function myPower(X, N) 
 abN = Abs(N): P = 1 
 Do While abN > 0 
   If (abN Mod 2) = 1 Then P = P * X 
   abN = abN \ 2: X = X * X 
 Loop 
 myPower = P 
 If N < 0 Then myPower = 1 / P 
End Function 

 

以上のことから，素直にプログラミングした例を示します。なお，論理的

には，連分数展開で求まった Eに対して，myPowerを使って， 

         myExpF = E * myPowor(A, K) 

  )12((2/(106/(2/(21e 222x n/xxxxx

 xx 2/(21e x

nxnx eee 

11  x 0x

1x

としても良さそうですが，E と AKの大きさが違いすぎるので，計算誤差が大

きくなります。 

 

【プログラム例 1】 
Public Function myExpF(X) 
 N = 100: XX = Abs(X): K = Fix(XX): XX = XX - K 
 A = XX * XX: E = 0 
 For i = N To 1 Step -1 
   E = A / (2 * (2 * i + 1) + E) 
 Next 
 E = 1 + 2 * XX / (2 - XX + E) 
 A = 2.71828182845905 
 Do While K > 0 
   If (K Mod 2) = 1 Then E = E * A 
   K = K \ 2: A = A * A 
 Loop 
 myExpF = E 
 If X < 0 Then myExpF = 1 / E 
End Function 

 

対数のところで示したように，e の値が近似値であることによる誤差が生

じます。対数の場合と同じ方法で改良します。 

 

【プログラム例 2】 
 
Public Function myExpF2(X) 

Alf = 0.523536028747135: P = Alf / (1 - Alf): PP = 1 + P 
 N = 100: XX = Abs(X): K = Fix(XX): XX = XX - K 
 A = XX * XX: E = 0 
 For i = N To 1 Step -1 
   E = A / (2 * (2 * i + 1) + E) 
 Next 
 E = 1 + 2 * XX / (2 - XX + E) 
 A1 = 2.71828182845904: A2 = 2.71828182845905 
 Do While K > 0 
   If (K Mod 2) = 1 Then E = E * (A1 + P * A2) / PP 
   K = K \ 2: A1 = A1 * A1: A2 = A2 * A2 
 Loop 
 myExpF2 = E 
 If X < 0 Then myExpF2 = 1 / E 

End Function 

 

以下に，myExpF と myExpF2 の比較を示します。当然のことながら

myExpF2 の方が良い結果となっています。 

 

 

 

 

 

 


